/*

Daniel Silverman	
Can Factual Misperceptions be Corrected? An Experiment on American Public Fears of Terrorism	


HYPOTHESES

Stated-Hyp1: Providing information regarding the risk of a terror attack will 
mitigate public fears of risk of terror. (p.1)

	Test-Hyp1: There will be a difference in percieved risk of a terror attack if information is provided about actual risk of terror vs. no information provided about risk.

Stated-Hyp2: Providing information regarding the risk of a terror attack will 
mitigate public fears of risk of terror if there is an elite cue. (p.5)

	Test-Hyp2: There will be a difference in percieved risk of a terror attack if information is provided about actual risk of terror vs. Democract elite cue is provided.

	Test-Hyp3: There will be a difference in percieved risk of a terror attack if information is provided about actual risk of terror vs. Republic elite cue is provided.	

	Test-Hyp4: There will be a difference in percieved risk of a terror attack if information is provided about actual risk of terror vs. military elite cue is provided.	
	
Stated-Hyp3: There will a difference in the effects of co-partisan, cross-partisan, and military elite cues (p. 5)

	Test-Hyp5: Within respondents who identify as partisan, there will be a difference in percieved risk of a terror attack if co-partisan elite cue vs. military elite cue is provided.

	Test-Hyp6: Within respondents who identify as partisan, there will be a difference in percieved risk of a terror attack if cross-partisan elite cue vs. military elite cue is provided.

	Test-Hyp7: Within respondents who identify as partisan, there will be a difference in percieved risk of a terror attack if co-partisan elite cue vs. cross-partisan elite cue is provided.	


Stated-Hyp4: The effect of co-partisan, cross-partisan, and military cues will vary across Democrats, Republicans, and Independents (p. 5)

	Test-Hyp8: Within co-partisan cue, Democrat and Republican respondents will vary.
	// cannot test for independents, because they have no copartisan elites
	
	Test-Hyp9: Within cross-partisan cue, Democrats and Republicans will vary.
	// cannot test for independents, because they have no cross-partisan elites
		
	Test-Hyp10: Within military elite cue, Democrats and Republicans will vary.
	Test-Hyp11: Within military elite cue, Democrats and Independents will vary.
	Test-Hyp12: Within military elite cue, Republicans and Independents will vary.	
	
********************************************************************************
* NOTES:

- there are a number of outcomes and pre-treatment measures; not sure how these 
relate to the experiment. In memo to reviewers, authors state that Q4 is what 
they're actually trying to measure and it is most relevant to their treatment. 
We are using Q4 for analysis. 
- restricting hypothesis 3 to partisans because military option is the only one 
open to independents, which can lead to confounding.
*/

clear all
use "Silverman1035.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

	tab P_TESS031

/*
1	Control group: No Corrective Information
2	Treatment 1: Corrective Information, No Elite Endorsement
3	Treatment 2: Corrective Information, Democratic Endorsement
4	Treatment 3: Corrective Information, Republican Endorsement
5	Treatment 4: Corrective Information, Military Endorsement
*/

* information only vs. control
	recode  P_TESS031 (1=0) (2=1) (*=.), gen(information)
	tab information
	
* elite type vs. information
	lab def elite 0 "0 info only" 1 "1 dem elite" 2 "2 rep elite" 3 "3 military elite"
	gen elite_type = P_TESS031 -2
	replace elite_type=. if P_TESS031 ==1 
	lab val elite_type elite
	tab elite_type, mis

* 	respondent party id
// NOTE: only applicable to respondents who identify as partisan
	tab PARTYID7
	
	* democrat vs. rep
	recode PARTYID7 (1/3=1) (5/7=0) (*=.), gen(resp_dem_rep)
	tab resp_dem_rep	
	
	* democrat vs. indep
	recode PARTYID7 (1/3=1) (4=0) (*=.), gen(resp_dem_indep)
	tab resp_dem_indep
	
	* republican vs. indep
	recode PARTYID7 (4=0) (5/7=1) (*=.), gen(resp_rep_indep)
	tab resp_rep_indep	
	
* co-partisan or cross-partisan cue?
	
	* cue aligned or not vs. military
	lab def alignment 0 "0 cross-partisan" 1 "1 co-partisan" 2 "2 military"
	gen alignment = .
	replace alignment = 1 if resp_dem_rep==1 & elite_type==1 // co party
	replace alignment = 1 if resp_dem_rep==0 & elite_type==2 // co party
	replace alignment = 0 if resp_dem_rep==1 & elite_type==2 // cross party
	replace alignment = 0 if resp_dem_rep==0 & elite_type==1 // cross party

	replace alignment = 2 if elite_type==3
		
	lab val alignment alignment
	tab alignment resp_dem_rep, mis

* OUTCOME MEASURES
	// NOTE: there are multiple post-treatment questions, but the authors explain in reviewer memo (page #2) that their intended outcome variable is personal risk as measured by question 4.
	
/*
Q4.  
How worried are you that you will become a victim of terrorism?

*/

	tab Q4
	// recode missing; and correct reverse coding
	replace Q4=. if Q4>4
	gen personalrisk= 5-Q4
	tab personalrisk, mis
	
********************************************************************************

* ANALYSIS
	

*Test-Hyp1: There will be a difference in percieved risk of a terror attack if information is provided about actual risk of terror vs. no information provided about risk.

	reg personalrisk i.information
	// reject. p=0.183 
	tess 1.information, init(Silverman1035)
	
*Test-Hyp2: There will be a difference in percieved risk of a terror attack if information is provided about actual risk of terror vs. Democract elite cue is provided.

	reg personalrisk i.elite_type if elite_type!=2&elite_type!=3
	// reject. p=0.418
	tess 1.elite_type
	
*Test-Hyp3: There will be a difference in percieved risk of a terror attack if information is provided about actual risk of terror vs. Republican elite cue is provided.	

	reg personalrisk i.elite_type if elite_type!=1&elite_type!=3
	// reject. p=0.214
	tess 2.elite_type
	
*Test-Hyp4: There will be a difference in percieved risk of a terror attack if information is provided about actual risk of terror vs. military elite cue is provided.	
	
	reg personalrisk i.elite_type if elite_type!=1&elite_type!=2
	// reject. 0.754
	tess 3.elite_type
	
*Test-Hyp5: Within respondents who identify as partisan, there will be a difference in percieved risk of a terror attack if co-partisan elite cue vs. military elite cue is provided.

	reg personalrisk i.alignment if resp_dem_rep!=. & alignment!=0
	// reject. 0.671
	tess 2.alignment
	
*Test-Hyp6: Within respondents who identify as partisan, there will be a difference in percieved risk of a terror attack if cross-partisan elite cue vs. military elite cue is provided.

	reg personalrisk i.alignment if resp_dem_rep!=. & alignment!=1
	// reject. 0.348
	tess 2.alignment
	
*Test-Hyp7: Within respondents who identify as partisan, there will be a difference in percieved risk of a terror attack if co-partisan elite cue vs. cross-partisan elite cue is provided.	

	reg personalrisk i.alignment if resp_dem_rep!=. & alignment!=2
	// reject. 0.142
	tess 1.alignment

*Test-Hyp8: Within co-partisan cue, Democrat and Republican respondents will vary.

	reg personalrisk i.resp_dem_rep if alignment==1
	// reject. 0.462
	tess 1.resp_dem_rep

*Test-Hyp9: Within cross-partisan cue, Democrats and Republicans will vary.
	
	reg personalrisk i.resp_dem_rep if alignment==0
	// reject. 0.193
	tess 1.resp_dem_rep	
	
*Test-Hyp10: Within military elite cue, Democrats and Republicans will vary.

	reg personalrisk i.resp_dem_rep if alignment==2
	// reject. 0.090
	tess 1.resp_dem_rep	

*Test-Hyp11: Within military elite cue, Democrats and Independents will vary.

	reg personalrisk i.resp_dem_indep if alignment==2
	// reject. 0.882
	tess 1.resp_dem_indep
	
*Test-Hyp12: Within military elite cue, Republicans and Independents will vary.	
	
	reg personalrisk i.resp_rep_indep if alignment==2
	// reject. 0.132
	tess 1.resp_rep_indep	
